

CH3_002

ANSWER

CLIENTGROUP

1 Dump Clients

Finished Code:

```
function dumpClientGroup()
{
    echo("Client Group (", ClientGroup.getID() ,
        ") contains these clients:");

    %count = ClientGroup.getCount();
    for (%i = 0; %i < %count; %i++)
    {
        %clientConn = ClientGroup.getObject(%i);
        echo("Client #", %i, " ID: ", %clientConn.getID() );
    }
}
```

Answers:

1. It would fail because there is no SimGroup named ClientGroup on instances of TGE that are in Client-Only mode.
2. Add the following code at the beginning of the function.

```
if(! isObject( ClientGroup ) ) return
```

CLIENTGROUP

2 Identifying The Local Client

Finished Code:

```
function dumpClientGroup2()
{
    echo("Client Group (", ClientGroup.getID() ,
        ") contains these clients:");

    %count = ClientGroup.getCount();
    for (%i = 0; %i < %count; %i++)
    {
        %clientConn = ClientGroup.getObject(%i);

        if( isObject( localClientConnection ) &&
            (%clientConn == localClientConnection.getID() ) )
        {
            echo("Client #", %i, " ID: ", %clientConn.getID() ,
                " \c2(localClientConnection)" );
        }
        else
        {
            echo("Client #", %i, " ID: ", %clientConn.getID() );
        }
    }
}
```

CLIENTGROUP

3 Getting The Server Connection

Finished Code:

```
function dumpClientGroup3()
{
    echo("Client Group (", ClientGroup.getID() ,
        ") contains these clients:");

    %count = ClientGroup.getCount();
    for (%i = 0; %i < %count; %i++)
    {
        %clientConn = ClientGroup.getObject(%i);

        if( isObject( localClientConnection ) &&
            (%clientConn == localClientConnection.getID() ) )
        {
            echo("Client #", %i, "\n > ClientConnection: ",
                %clientConn.getID() , " \c2(localClientConnection)" );
        }
        else
        {
            echo("Client #", %i, "\n > ClientConnection: ",
                %clientConn.getID() );
        }

        %serverConn = %clientConn.getServerConnection();

        echo(" > ServerConnection: ", %serverConn );
    }
}
```