

CH3_003

ANSWER

GHOST RESOLUTION

1 Counting Ghosts

Finished Code:

```
function dumpClientGroup4()
{
    echo("Client Group (", ClientGroup.getID() ,
        ") contains these clients:");

    %count = ClientGroup.getCount();
    for (%i = 0; %i < %count; %i++)
    {
        %clientConn = ClientGroup.getObject(%i);

        if( isObject( localClientConnection ) &&
            (%clientConn == localClientConnection.getID() ) )
        {
            echo("Client #", %i, "\n > ClientConnection: ",
                %clientConn.getID() , " \c2(localClientConnection)" );
        }
        else
        {
            echo("Client #", %i, "\n > ClientConnection: ",
                %clientConn.getID() );
        }

        %serverConn = %clientConn.getServerConnection();

        echo(" > ServerConnection: ", %serverConn );
        echo(" >    Active Ghosts: ", %serverConn.getGhostsActive() );
    }
}
```

GHOST RESOLUTION

Answers:

1. The five server objects that are being ghosted to the local client are:

- Red AIPlayer
- Green AIPlayer
- Blue AIPlayer
- The Player Object (also the control object)
- The Camera

2 Dumping Ghost Indexes

Finished Code:

```
function dumpLocalGhostIndexes()  
{  
    %cameraID = localClientConnection.camera;  
    %playerID = localClientConnection.player;  
  
    %cameraGhost = localClientConnection.getGhostID( %cameraID );  
    %playerGhost = localClientConnection.getGhostID( %playerID );  
  
    echo("The server object ID of the local camera is: ", %cameraID );  
    echo("The server object ID of the local player is: ", %playerID );  
  
    echo("The ghost index for the local camera is: ", %cameraGhost );  
    echo("The ghost index for the local player is: ", %playerGhost );  
  
}
```

GHOST RESOLUTION

3 Resolve Server Object IDs

Finished Code:

```
function dumpServerObjectIDs()
{
    %ghostCount = serverConnection.getGhostsActive();

    echo("Resolving ghost indexes to server object IDs.");

    for( %count = 0; %count < %ghostCount; %count++ )
    {
        %serverObjectID = serverConnection.resolveGhostID( %count );
        echo("Ghost Index #", %count, " translates to server object ID: ",
            %serverObjectID ,
            " Class Name: ", %serverObjectID.getClassName());
    }
}
```

4 Resolve Ghost Object IDs

Finished Code:

```
function dumpGhostIDs()
{
    %ghostCount = serverConnection.getGhostsActive();

    echo("Resolving ghost indexes to ghosts for local server and client.");

    for( %count = 0; %count < %ghostCount; %count++ )
    {
        %ghostID =
            localClientConnection.resolveObjectFromGhostIndex(%count);
        echo("Ghost Index #", %count, " translates to ghost object ID: ",
            %ghostID ,
            " Class Name: ", %ghostID.getClassName());
    }
}
```